

“十五”国家重点电子出版物规划项目·计算机知识普及和软件开发系列
专业程序员编程实作丛书 (11)

Visual Basic 6.0

Programming
Instances

全实例 教程


北京希望电子出版社 总策划
李真文 编写

本版特点

- * 使用 105 个经典实例讲解 VB 编程的技巧与经验

适用范围

- * 面向初中级编程人员
- * 高校相关专业师生
- * 社会相关培训班学员



北京希望电子出版社
Beijing Hope Electronic Press
www.bhp.com.cn

“十五”国家重点电子出版物规划项目·计算机知识普及和软件开发系列
专业程序员编程实作丛书(11)

Visual Basic 6.0

Programming
Instances

全实例 教程



B1233162

北京希望电子出版社 总策划
李真文 编写

本版特点

* 使用105个经典实例讲解VB编程的技巧与经验

适用范围

- * 面向初中级编程人员
- * 高校相关专业师生
- * 社会相关培训班学员



北京希望电子出版社
Beijing Hope Electronic P
www.bhp.com

内 容 简 介

本书使用了 105 个经典实例讲解了 VB 编程的技巧与经验。

本书由 8 章组成。内容包括：第一章窗体，介绍各种控制窗体的方法与技巧；第二章控件，用生动范例介绍了各种控件的使用；第三章多媒体，介绍了一些编写游戏和动画的实例；第四章图像和光标，介绍图像显示的方法与应用技巧；第五章 Windows API，介绍 API 控制的各种编程技巧；第六章介绍数据库编程及数据库应用技巧；第七章网络，介绍用 VB 编写网络应用程序；第八章介绍一些用 VB 编写的常用小程序。

本书内容丰富，所有范例都是功能介绍与代码分析相结合，并给出源代码。是 VB 程序开发人员的最佳参考书，也可作为高等院校相关专业师生的教学参考书。

本版 CD 内容为实例源代码。

读者在本书使用过程中的技术问题请与作者联系。E-mail: dotnetcomponent@Yahoo.com.cn。

系 列 盘 书： “十五”国家重点电子出版物规划项目 计算机知识普及和软件开发系列
专业程序员编程实作丛书（11）

盘 书 名： Visual Basic 6.0 全实例教程

总 策 划： 北京希望电子出版社

文 本 著 作 者： 李真文 编写

C D 制 作 者： 希望多媒体开发中心

C D 测 试 者： 希望多媒体测试部

责 任 编 辑： 周凤明

出 版、发 行 者： 北京希望电子出版社

地 址： 北京海淀区知春路 63 号卫星大厦三层，100080

网址：www.bhp.com.cn E-mail: lwm@hope.com.cn

电 话： 010-62520290,62521724,62528991,62630301,62524940,62521921,82610344

（发行） 010-62613322-215（门市） 010-82675588-501,82675588-201（编辑部）

经 销： 各地新华书店、软件连锁店

排 版： 希望图书输出中心 马君

C D 生 产 者： 北京中新联光盘有限责任公司

文 本 印 刷 者： 北京媛明印刷厂

开 本 / 规 格： 787 毫米×1092 毫米 16 开本 21 印张 477 千字

版 次 / 印 次： 2002 年 6 月第 1 版 2002 年 6 月第 1 次印刷

印 数： 1-3000 册

本 版 号： ISBN 7-900101-06-3

定 价： 33.00 元（本版 CD）

说明：凡我社产品如有残缺，可执相关凭证与本社调换。

前 言

由于 Basic 语言本身具有非常容易上手的特点, 以及 Visual Basic 最彻底、紧密地与 Windows 平台相结合, 在 Windows 平台上, 采用 Visual Basic 开发的软件在数量上无疑是最多的一种类型之一。所以, 搜集并研究 Visual Basic 典型应用的解决方案显得非常必要, 这使得最大多数的 Visual Basic 程序员在做任何项目时, 可以参考前人在相关领域已经开展的工作, 很有可能一些花费数小时、数天甚至数周冥思苦想而不得解决的问题, 在顷刻之间找到了答案, 这正是本书的编写目的。

本书搜集并分析了 105 个包含完整实现的 Visual Basic 程序代码, 这些程序代码涉及 6 个 Visual Basic 应用领域: 窗体、控件、游戏编程、图像和光标、WINDOWS API 和数据库。其中, 窗体领域涉及的应用包括在窗体上显示橡皮带式生成线、非常酷的不规则图形窗体、图片覆盖窗体的效果、三种动态窗体的显示、限定窗体的最大和最小尺寸、模拟 DOS 窗体、自动调整窗体中控件的尺寸、定制自己的窗体、活动窗体、雷达效果的窗体、停靠窗体、滚动字幕的窗体、可以换皮肤的窗体、半透明窗体和几种关闭窗口体的方式、对窗体样式的控制、显示每日一帖; 控件领域涉及的应用包括漂亮的滑块控件、定制的几种常用控件、系统托盘控件、渐变色的进度条控件、非常漂亮的 List 列表、封装的 WinSock 控件、定时更改桌面图片、漂亮的按钮、立体文字输入框、定制的文件控件、用 Listbox 显示系统字体的几种形式、几种常用控件的特殊用法、3D 标签控件、扑克牌控件、不同颜色项的下拉框控件、液晶数字控件、功能强大的按钮控件、图形滑块控件; 游戏编程领域涉及的应用包括俄罗斯方块游戏、飞镖游戏、游蛇游戏、数字方块游戏、基于 DirectDraw 的游蛇游戏、冒险游戏、滚轴 RPG 游戏、非常好玩的小蜜蜂游戏、射击游戏、井字棋游戏; 图像和光标领域涉及的应用包括从资源文件中加载不同格式的图片文件、图标的几种常用的显示方式、显示透明位图、取屏幕上的任意点的信息、操作 VB 图形的颜色、简单的位图处理、在背景图上平滑的移动位图、在程序中使用动画光标、产生炫目的波纹效果图、模拟 DNA 效果图; WINDOWS API 领域涉及的应用包括获取 Windows 系统时间和运行时间、获取电源状态信息、锁定鼠标、获取磁盘卷信息、自动切换输入法、用 API 函数调用打开文件对话框、机器的 IP 地址、打开文件属性对话框、任意角度、大小显示文字、显示开始按钮、给菜单加上图片、执行和关闭指定的程序、画双列的菜单、自制窗口管理器、任务栏图标的实现、注销用户和重启、关闭计算机; 数据库领域涉及的应用包括把数据库转换为文本文件、打印数据库记录、把数据库记录发送到 WORD2000 和 EXCEL2000、ACCESS 数据库小工具、压缩数据库、修复数据库、保存 ADO 数据库为 XML 格式数据库、查询小工具、数据库转换工具、直接打印 Microsoft Access 报表。

本书由编程高手工作室集体完成。其中, 李真文负责丛书策划和组织, 刘晓军负责文档编写和程序测试的管理工作, 熊国强负责本书的编写。另外, 下列人员参与了本书的编写工作: 王锐、周杰、曾岚鸢、李斌、陈兵、刘晖、程杜平、刘湘宁、吕秀娟、张景新、王伟民、刘堃、邢丽丽、韩春雷、雷军、仇美娟、郭月强、黄聪、许敏、邹秋玲、卢绍青、王海滨、王福建、张洁、张奇支、张津、任伟、王炜、武焰、许传斌、周冠卿、范蕾、谢

宏涛、汤华、方树财、马米、杜平周、孙才、吴爱兵、亚敏、谢宏韬、田林燕、刘鹏、马全明、成勇、张宝平、魏初瞬、赵巍、高耘、葛英、桂蓉、郭建梅、张翠平、温智旻、王振祥、刘晓宇、李晓冬、朱继莲、朱卫江、张炬、钟伦、王朝霞、石少强。由于学识疏浅，书中一定有不尽人意的地方，敬请读者批评指正，Email: code.expert@263.net。其中，邮件标题需要给出丛书的具体图书标题，邮件内容需要说明部分、章和程序的编号，以及详细的 bug 报告，以便有我们的技术支持人员有针对性地解决你的问题。

光盘使用说明

- ◆ 本书配套光盘包含所有程序的完整文件，并在 Visual Basic 编译、运行通过，准确的编译、运行环境需要阅读具体的程序所在的章的介绍。
- ◆ 根目录下的目录的数字编号对应图书的章的编号，而章里的程序编号与章的目录下的目录编号一致。在查找图书的程序源码时，请以此为参考。
- ◆ 在使用程序之前，需要手工拷贝该程序的所有文件和目录到硬盘的适当位置。需要特别说明的是，在用 Visual Basic 开打之前，需要去掉所有文件和目录的只读属性。

如果有任何问题，请 Email: code.expert@263.net。其中，邮件标题需要给出丛书的具体图书标题，邮件内容需要说明部分、章和程序的编号，以及详细的 bug 报告，以便有我们的技术支持人员有针对性地解决你的问题。

目 录

第1章 窗体	1
程序 1 在窗体上显示橡皮带式生成线	1
程序 2 非常酷的不规则图形窗体	3
程序 3 图片覆盖窗体的效果	6
程序 4 三种动态窗体的显示	7
程序 5 限定窗体的最大和最小尺寸	9
程序 6 模拟 DOS 窗体	11
程序 7 自动调整窗体中控件的尺寸	14
程序 8 定制自己的窗体	17
程序 9 活动窗体	18
程序 10 雷达效果的窗体	20
程序 11 停靠窗体	21
程序 12 滚动字幕的窗体	24
程序 13 可以换皮肤的窗体	25
程序 14 半透明窗体和几种关闭窗体的方式	29
程序 15 对窗体样式的控制	31
程序 16 显示每日一帖	33
第2章 控件	35
程序 1 漂亮的滑块控件	35
程序 2 定制的几种常用控件	37
程序 3 系统托盘控件	41
程序 4 渐变色的进度条控件	43
程序 5 非常漂亮的 List 列表	45
程序 6 封装的 WinSock 控件	46
程序 7 定时更改桌面图片	48
程序 8 漂亮的按钮	50
程序 9 立体文字输入框	51
程序 10 定制的文件控件	53
程序 11 用 listbox 显示系统字体的几种形式	56
程序 12 几种常用控件的特殊用法	58
程序 13 3D 标签控件	62
程序 14 扑克牌控件	64
程序 15 不同颜色项的下拉框控件	66
程序 16 液晶数字控件	68
程序 17 功能强大的按钮控件	69
程序 18 图形滑块控件	72

第 3 章 多媒体	74
程序 1 俄罗斯方块游戏.....	74
程序 2 飞镖游戏.....	77
程序 3 游蛇游戏.....	79
程序 4 数字方块游戏.....	81
程序 5 基于 DirectDraw 的游蛇游戏	83
程序 6 冒险游戏.....	85
程序 7 滚轴 RPG 游戏.....	88
程序 8 非常好玩的小蜜蜂游戏.....	92
程序 9 射击游戏.....	98
程序 10 井字棋游戏.....	100
程序 11 发声器.....	102
程序 12 弹动的方块.....	106
程序 13 旋转的陀螺.....	112
程序 14 刷新图片的效果.....	114
程序 15 在背景图上平滑的移动位图	117
第 4 章 图像和光标	119
程序 1 从资源文件中加载不同格式的图片文件	119
程序 2 图标的几种常用显示方式.....	121
程序 3 显示透明位图.....	124
程序 4 获取屏幕上任意点的信息.....	127
程序 5 操作 VB 图形的颜色	129
程序 6 简单的位图处理.....	130
程序 7 在程序中使用动画光标.....	132
程序 8 产生炫目的波纹效果图.....	133
程序 9 模拟 DNA 效果图	134
程序 10 简单画图工具.....	136
程序 11 画轴式打开窗体方式.....	138
第 5 章 Windows API	140
程序 1 获取 Windows 系统时间和运行时间	140
程序 2 获取电源状态信息.....	142
程序 3 锁定鼠标.....	144
程序 4 获取磁盘卷信息.....	145
程序 5 自动切换输入法.....	148
程序 6 用 API 函数调用打开文件对话框.....	150
程序 7 打开文件属性对话框.....	153
程序 8 任意角度、大小显示文字.....	156
程序 9 隐藏、显示开始按钮.....	158
程序 10 给菜单加上图片.....	160

程序 11	执行和关闭指定的程序	162
程序 12	画双列的菜单	167
程序 13	自制窗口管理器	169
程序 14	任务栏图标的实现	174
程序 15	注销用户和重启、关闭计算机	175
程序 16	光驱的开关控制	179
第 6 章	数据库	181
程序 1	把数据库转换为文本文件	181
程序 2	打印数据库记录	183
程序 3	把数据库记录发送到 Word 2000 和 Excel 2000	186
程序 4	Access 数据库小工具	189
程序 5	压缩数据库	201
程序 6	修复数据库	202
程序 7	保存 ADO 数据库为 XML 格式数据库	204
程序 8	查询小工具	206
程序 9	数据库转换工具	211
程序 10	直接打印 Microsoft Access 报表	214
程序 11	把图片放到数据库中	216
程序 12	数据库的查询	223
程序 13	数据库转换 HTML	224
第 7 章	网络	233
程序 1	机器的 IP 地址	233
程序 2	网上连接	236
程序 3	用 VB6.0 编写的 Ping 软件	237
程序 4	显示本机登陆名以及想要登陆的网站	245
程序 5	FTP 传输	247
第 8 章	其他	252
程序 1	全球通用取款机	252
程序 2	修改密码	257
程序 3	一个文本编辑器	267
程序 4	获取 CPU 的报告	277
程序 5	检查驱动信息	286
程序 6	计算星期几	292
程序 7	获取文件信息	294
程序 8	进行格式化	299
程序 9	对文件进行 Split	301
程序 10	使用 INI 文件	308
程序 11	注册表实例	311

第 1 章 窗 体

程序 1 在窗体上显示橡皮带式生成线

功能介绍

本程序能够通过鼠标选择，在窗体上显示橡皮带式生成线，当松开鼠标时，返回一个矩形结构。此程序可以用于区域选择和屏幕区域捕捉，而且做到了在 VB 下实现动态选择矩形区域而不闪烁的效果，参见图 1.1。




图 1.1 在窗体上显示鼠标带式生成线

这个程序选择的有效区域只能在自己的 Form 窗体或者 VB 的编辑界面上，当然也可以参照其它 API 函数设置扩展到全部窗体，在返回矩形区域的同时，可以通过 API 函数将矩形区域转变成位图或者其他格式的图片。

鼠标松开时，程序返回一个矩形结构，参见图 1.2。




图 1.2 松开鼠标的时候返回一个矩形区域

代码分析

此程序的关键代码包含在 modRubberBand.bas 文件中；API 函数声明以及常量定义包含在 modAPIDeclares.bas 文件中；Rubberband\modCommon.bas 主要是进行错误处理，主要函数的功能及作用参见表 1.1。

表 1.1 关键函数列表

CaptureRect	此函数允许用户选择屏幕的一部分，当鼠标左键被释放时，屏幕中被选择的区域转换成一个位图，然后返回一个图片对象
DrawSelect	此函数绘制所选设备尺寸的矩形，即图 1.1 看到的矩形框
NormalizeRect	如果这个矩形设备是颠倒的，则交换它们。主要用于保证矩形坐标第一个在左上，第二个在右下

在 frm_RubberBand.frm 中的 Form_MouseDown 事件中调用绘制矩形区域的橡皮带式线的函数，参见程序清单 1.1。

程序清单 1.1 绘制矩形区域的橡皮带式线

```
Public Function CaptureRect (ByVal hWnd As Long, ptOrigin As POINTAPI)
As RECT
    Dim hDC As Long
    Dim msg As msg
    Dim hPal, hBitmap As Long
    Dim rcClip As RECT
    Dim rcClient As RECT
    ' 在这里取得屏幕关联
    hDC = CreateDCNull("DISPLAY", 0, 0, 0&)
    ' 转换屏幕关联
    ClientToScreen hWnd, ptOrigin
    ' rcClip 是用户当前选择的区域
    ' 当用户按下鼠标左键，初始化的一个非常小的区域
    rcClip.Left = ptOrigin.x
    rcClip.Right = ptOrigin.x
    rcClip.Top = ptOrigin.y
    rcClip.Bottom = ptOrigin.y
    ' 该函数能够使矩形区坐标为第一个坐标，该坐标在左上角
    NormalizeRect rcClip
    ' 开始绘画选择的区域带
    DrawSelect hDC, rcClip
    ' 循环取得鼠标的当前消息，一直到鼠标左键弹起事件被触发
    ' 其间一直画橡皮带式生成线，包括尺寸
```

```

Do
    ' 等待鼠标消息
    WaitMessage
    If (PeekMessage(msg, 0, WM_MOUSEFIRST, _
        WM_MOUSELAST, PM_REMOVE)) Then
        ' 擦去当前的矩形框
        DrawSelect hDC, rcClip
        ' 得到新的信息数据
        rcClip.Left = ptOrigin.x
        rcClip.Top = ptOrigin.y
        rcClip.Right = msg.pt.x
        rcClip.Bottom = msg.pt.y
        ' 验证是否到了末端
        NormalizeRect rcClip
        ' 画一个新的矩形
        DrawSelect hDC, rcClip
        If (msg.message = WM_LBUTTONDOWN) Then Exit Do
    End If
Loop
' 擦去最后一个区域信息
DrawSelect hDC, rcClip
' 清除设备
DeleteDC hDC
' 释放捕获
ReleaseCapture
' 得到选中的区域
CaptureRect = rcClip
' 传递鼠标弹起信息给窗体
PostMessage hWnd, WM_LBUTTONDOWN, msg.wParam, msg.lParam
End Function

```

源码路径: 01\01。

程序 2 非常酷的不规则图形窗体

功能介绍

本程序能够生成一个非常酷的不规则图形窗体，而且窗体本身就是由图片的不透明区域组成，同样响应鼠标事件，效果可参见图 1.3。




图 1.3 图形窗体效果演示

这个程序的原理是逐行读取像素，所以通过设定不同的透明颜色，可以产生各种窗体效果，如镂空效果等。

代码分析

本程序使用到的函数包含在模块文件 Module1.bas 中，参见表 1.2。

表 1.2 Module1.bas 中的函数列表

函数	说明
GetPixel	API 函数：在指定的设备场景中取得一个像素的 RGB 值
SetWindowRgn	API 函数：如果执行成功则根据参数改变窗体的区域
CreateRectRgn	API 函数：创建一个由点 X1, Y1 和 X2, Y2 描述的矩形区域
CombineRgn	API 函数：将两个区域组合为一个新区域
SendMessage	API 函数：将一条消息发给指定的窗体，直到消息处理完毕后才从窗体返回
ReleaseCapture	API 函数：为当前的应用程序释放鼠标捕获
DeleteObject	API 函数：删除指定的 GDI 对象，释放对象用到的所有系统资源
MakeRegion	根据一个给定的图片绘制窗体区域的函数

本程序主要使用 MakeRegion 函数，其原理是：逐行读取像素的颜色，判断取得的颜色和指定要透明的颜色是否相同，如果是，则为透明像素，否则为不透明像素。然后根据该行中不透明像素组成的区域，使用 API 函数 CreateRectRgn 创建一个矩形区域，反复使用 API 函数 CombineRgn，将新生成的区域和原来已经存在的区域合并，最后获得一个非常酷的不规则图形窗体，该函数的代码参见程序清单 1.2。

程序清单 1.2 根据一个给定的图片绘制窗体区域的函数

```
Public Function MakeRegion(picSkin As PictureBox) As Long
    ' 根据一个给定的 Picture Box 的图片创造窗体区域，逐行依次用不透明像素
    创建一个区域。
```

```

Dim X As Long, Y As Long, StartLineX As Long
Dim FullRegion As Long, LineRegion As Long
Dim TransparentColor As Long
Dim InFirstRegion As Boolean ' 是否是第一次绘制
Dim InLine As Boolean ' 标记是否是不透明的像素区域
Dim hDC As Long
Dim PicWidth As Long
Dim PicHeight As Long
hDC = picSkin.hDC
PicWidth = picSkin.ScaleWidth
PicHeight = picSkin.ScaleHeight
InFirstRegion = True: InLine = False
X = Y = StartLineX = 0
' 设置透明颜色:
' 这里用图片的第一个像素的颜色作要透明的颜色, 当然也可以用别的
' 颜色, 例如 TransparentColor=RGB(255,255,0)
TransparentColor = GetPixel(hDC, 0, 0)
For Y = 0 To PicHeight - 1
    For X = 0 To PicWidth - 1
        If GetPixel(hDC, X, Y) = TransparentColor Or X = PicWidth Then
            ' 如果到了要透明的行就绘制该行
            If InLine Then
                InLine = False
                LineRegion = CreateRectRgn(StartLineX, Y, X, Y + 1)
                If InFirstRegion Then
                    FullRegion = LineRegion
                    InFirstRegion = False
                Else
                    CombineRgn FullRegion, FullRegion, LineRegion, RGN_OR
                    DeleteObject LineRegion
                End If
            End If
        End If
    Else
        ' 如果到了不透明的行
        If Not InLine Then
            InLine = True
            StartLineX = X
        End If
    End If
End For

```

```

Next
Next
MakeRegion = FullRegion
End Function

```

这里还有一个很重要的地方，就是在 form1.frm 中的 picMainSkin_MouseDown 事件中，当鼠标在图片上拖动时，首先要释放 PictureBox 对鼠标事件的捕捉，然后发送一个鼠标拖动的消息给窗体，这里所说的窗体是一个看不见的窗体，这样，拖动图片的同时也拖动了窗体，参见程序清单 1.3。

程序清单 1.3 发送消息给窗体

```

Private Sub picMainSkin_MouseDown(Button As Integer, Shift As Integer, _
    X As Single, Y As Single)
    ReleaseCapture
    SendMessage Me.hWnd, WM_NCLBUTTONDOWN, HTCAPTION, 0&
End Sub

```

源码路径：01\02。

程序 3 图片覆盖窗体的效果

功能介绍

这个程序演示了将 PictureBox 图片覆盖到窗体上产生的特殊效果，这种效果可以用于程序载入时的 LOGO 窗体，也可以用于 ABOUT 对话框，参见图 1.4。




图 1.4 覆盖过程中的效果

代码分析

本程序的原理十分简单，炫目的效果全部用画线语句产生，产生此效果的函数 Zrak 包含在 FrmIntro.frm 文件中。在 Zrak 函数中，从 PictureBox 左边开始，循环从上至下取 PictureBox 图片的像素，然后，以 PictureBox 的右下角为起点，以取得像素的坐标为终点，用取得像素的颜色画一条直线，这样就产生了上图演示的效果，其中使用到的函数参见表 1.3。

表 1.3 程序使用的主要函数列表

函数	说明
Zrak	循环取 PictureBox 图片的像素画线
DoEvents	VB 自带函数：转让控制权，让操作系统处理其它事件

循环选取 PictureBox 图片的像素画线的函数参见程序清单 1.4。

程序清单 1.4 循环选取 PictureBox 图片的像素画线的函数

```

Private Function Zrak(slika As PictureBox, StartX As Integer, StartY
As Integer, Levo As
Integer, Desno As Integer, Boja As String)
Me.ScaleMode = vbPixels
With slika
    .ScaleMode = vbPixels
    .AutoRedraw = True
End With
For Proba2 = 0 To slika.ScaleWidth
    DoEvents ' 转让控制权，以便让操作系统处理其它的事件。
    For Proba = 0 To slika.ScaleHeight
        Boja2 = slika.Point(Proba2, Proba)
        Line (StartX, StartY)-(Levo + Proba2, Desno + Proba), Boja2
    Next
    Line (StartX, StartY)-(Levo + Proba2, Desno + slika.ScaleHeight), Boja2
Next
For Proba2 = 0 To slika.ScaleHeight
    Line (StartX, StartY)-(Levo + slika.ScaleWidth, Desno + Proba2), Boja2
Next
Timer1.Interval = 0
End Function

```

这里要注意的是，在循环取 PictureBox 图片的像素画线的函数中，使用 DoEvents() 函数在画线过程中转让控制权，让操作系统处理其它事件。

源码路径：01/03。

程序 4 三种动态窗体的显示

功能介绍

这个程序演示了三种动态窗体的显示效果，分别为从左上角弹出、由中心向边缘扩展、渐隐显现，其中渐隐显现效果必须在 Windows 2000 下才能显示正常，程序运行界面参见图 1.5。




图 1.5 程序运行时的界面

代码分析

本程序的关键代码包含在模板 Module1.bas 文件中,其中主要函数以及说明参见表 1.4。

表 1.4 Module1.bas 中主要函数列表

函数	说明
AnimateWindow	API 函数: 在显示和隐藏窗体的时候可以使用特殊效果
CopyMemory	API 函数: 拷贝一块内存区到另外一个区域
GetWindowLong	API 函数: 从指定窗体的结构中取得信息
SetWindowLong	API 函数: 在窗体结构中为指定的窗体设置信息
GetProp	API 函数: 从指定窗体的属性列表中返回一个属性
SetProp	API 函数: 添加和改变指定窗体属性列表中的一个属性
RemoveProp	API 函数: 从指定窗体的属性列表中移去一个属性
CallWindowProc	API 函数: 通过一个消息给指定的窗体
OleTranslateColor	API 函数: 转变 OLE_COLOR 类型到 COLORREF 类型
TranslateColor	API 函数: 使指定的颜色透明
AnimWndProc	调用 CallWindowProc 发送 WM_PRINTCLIENT 消息给窗体, 绘制内容区
SubclassAnim	添加窗体的属性, 使之初始化时调用 AnimWndProc 函数产生动态效果
UnSubclassAnim	关闭窗体时, 移取窗体添加的属性

窗体弹出的三种效果都是通过 Form1.frm 中三个按钮的点击事件调用 API 函数 AnimateWindow 实现的。为了让子窗体能够用指定的效果显示,要调用 SubclassAnim 函数做一个初始化工作,参见程序清单 1.5。

程序清单 1.5 子程序初始化时调用的函数

```
Public Sub SubclassAnim(frm As frmAnim)
    Dim l As Long
    If GetProp(frm.hWnd, "ExAnimWndProc") <> 0 Then
        'Already subclassed      '已归入子程序
        Exit Sub
    End If
    l = GetWindowLong(frm.hWnd, GWL_WNDPROC)
    SetProp frm.hWnd, "ExAnimWndProc", l
End Sub
```